How to have a transparent-like kind of Delphi Form

by Francois Gaillard - f_gaillard@compuserve.com 

Introduction

I was searching the Win32 API for a desktop procedure when I saw a PaintDesktop routine in the list. As I'm curious I moved to the help page, where it is described as:

"The PaintDesktop function fills the clipping region in the specified device context with the desktop pattern or wallpaper."

Wow! That gave me the idea to build Forms that could show the desktop image as an automatically painted background. This would produce a clean sort of transparent effect: not showing any icon or underlying window, and solid enough so that you click on it without passing through.

Simple SDI Form

A direct implementation

I made a procedure to repaint the background of the Form passed as argument. Using the provided canvas avoids caring of the Device Context.

procedure PaintWithDesktop(AForm: TForm);
{ 1st try : straightforward }
begin
  with AForm do
    PaintDeskTop(Canvas.handle)
end;
So you just call it from a button to see what happens, and it works. You see your desktop image behind your TWinControls (Edit, Buttons, …). But if you had any TLabel, TImage or Tshape, they disappeared!

Keeping the TGraphicControls visible

What happened? You put another bitmap above the background in the client area of the Form. Remember that a TGraphicControl do not own any canvas but use the one of the parent. Thus they are painted on the background, and always shown under any TWinControl. Those ones, having their own canvas are not affected.

Therefore all the TGraphicControls are to be repainted after the PaintDesktop. This is done by a WM_PAINT message addressed to each of these ones, found in the Controls property. It's better to use a Perform then a SendMessage because we don't need to send it outside.

There is a special (one could say dirty) trick to get the canvas of a TGraphicControl in a generic manner, even if it has been hidden by being protected.

If Canvas had been public, writing "(AnyDescendant as TGraphicControl).Canvas" would have been legal. And a lot of  TGraphicControls, like the TShape, do not even surface their canvas! A very simple workaround to get that kind of "inherited Canvas" is to cast with the simplest available descendant class which makes its canvas public.

procedure PaintWithDesktop(AForm: TForm);
{ 2nd try : repainting Graphic controls }
var
  i: integer;
begin
  with AForm do begin
    { Let's surface the desktop image }
    PaintDeskTop(Canvas.handle);
    { ask for a repaint of all Graphic controls}
    for i:=0 to ControlCount-1 do
      if Controls[i] is TGraphicControl then
        { ******    SPECIAL TRICK : USE WITH CARE    ******     }
        { As Canvas is a protected property in TGraphicControl, }
        { we'll use the simplest descendant to surface it       }
        Controls[i].perform(WM_PAINT, TLabel(Controls[i]).canvas.handle, 0);
  end; { with }
end;
I used TLabel but you could create a special class :

type

  TCanvasGraphicControl = class(TGraphicControl)

  public

    property Canvas;

  end;

... 

Controls[i].perform(WM_PAINT, TCanvasGraphicControl(Controls[i]).canvas.handle, 0);
We need some automation

To keep our desktop visible, we need to repaint the desktop on the Form each time it has been redrawn. So we have to intercept the WM_PAINT message and place a PaintWithDesktop routine after the normal painting.

But it is not enough! Try to move your Form, and you'll see that piece of desktop image moving here and there.

We have to intercept also the WM_MOVE message.

In the private declarations of your Form, place the messages handlers.
private
    procedure WMMove(var Message: TWMMove); message WM_MOVE;
    procedure WMPaint(var Message: TWMPaint); message WM_PAINT;

And implement them below.

procedure TForm1.WMMove(var Message: TWMMove);
begin
  inherited;
  PaintWithDesktop(self);
end;

procedure TForm1.WMPaint(var Message: TWMPaint);
begin
  inherited;
  PaintWithDesktop(self);
end;
MDI Form

If there is a place where to show the desktop seems obviously useful, it is in the main form of an MDI application. It is generally just a work frame for the children windows.

Main MDI Form : same as SDI?

If we try the same technique, there seems to be a problem. We do not see any desktop painting. Same thing happens with any TgraphicControl dropped onto a main MDI form and remaining hidden.

It's because of the MDI Client Window which occupies the client area of an MDI form and serves as the background for MDI child windows, providing support for creating and manipulating child windows.

So we need to deal with MDI client and paint our desktop onto it too.

procedure PaintWithDesktop(AForm: TForm);
{ repainting Graphic controls and handling(?) MDI }
var
  h: HWND;
  i: integer;
begin
  with AForm do begin
    { simply paint Desktop on Canvas }
    PaintDeskTop(Canvas.handle);
    if FormStyle = fsMDIForm then begin
      { need to deal also with Client Window to paint Desktop }
      h:=getDC(ClientHandle);
      if h<>0 then begin
        PaintDesktop(h);
        ReleaseDC(ClientHandle,h);
      end;
    end; { if FormStyle... }
    { ask for a repaint of all Graphic controls}
    for i:=0 to ControlCount-1 do
      if (Controls[i] is TGraphicControl) and Controls[i].HasParent then begin
        { ******    SPECIAL TRICK : USE WITH CARE    ******     }
        { As Canvas is a protected property in TGraphicControl, }
        { we'll use the simplest descendant to surface it       }
        Controls[i].perform(WM_PAINT, TLabel(Controls[i]).canvas.handle, 0);
      end;
  end; { with }
end;


Well it begins to look great. Let's put some MDI Child onto it.

Child MDI Form : same as SDI

Of course we may want also to paint our destop onto the MDI children.

The very same technique that we used for SDI forms works also with MDI children.

{---------------------------------------------------------}
    procedure WMPaint(var Message: TWMPaint); message WM_PAINT;
    procedure WMMove(var Message: TWMMove); message WM_MOVE;
{---------------------------------------------------------}
...
{---------------------------------------------------------------}
procedure TMDIChild.WMPaint(var Message: TWMPaint);
 begin
  inherited;
  PaintWithDesktop(self);
end;

procedure TMDIChild.WMMove(var Message: TWMMove);
begin
  inherited;
  PaintWithDesktop3(self);
end;
{---------------------------------------------------------------}
Great! Looks like what we want.

Back to the Main MDI Form

But try to move your main MDI window… MDI children are stuck with their snapshot portion of desktop.

We obviously need to repaint that desktop at the same time we paint it onto the main form.

procedure PaintWithDesktop(AForm: TForm);
{ last try (?) : repainting Graphic controls and handling MDI }
var
  h: HWND;
  i: integer;
begin
  with AForm do begin
    { simply paint Desktop on Canvas }
    PaintDeskTop(Canvas.handle);
    if FormStyle = fsMDIForm then begin
      { need to deal also with Client Window to paint Desktop }
      h:=getDC(ClientHandle);
      if h<>0 then begin
        PaintDesktop(h);
        ReleaseDC(ClientHandle,h);
        { repaint also the MDI Children Forms }
        for i:= MDIChildCount-1 downto 0 do
          PaintWithDesktop(MDIChildren[i]);
      end;
    end; { if FormStyle... }
    { ask for a repaint of all Graphic controls}
    for i:=0 to ControlCount-1 do
      if (Controls[i] is TGraphicControl) and Controls[i].HasParent then begin
        { ******    SPECIAL TRICK : USE WITH CARE    ******     }
        { As Canvas is a protected property in TGraphicControl, }
        { we'll use the simplest descendant to surface it       }
        Controls[i].perform(WM_PAINT, TLabel(Controls[i]).canvas.handle, 0);
      end;
  end; { with }
end;

Much better! But if we move the child window… The client window is repainted with it's solid color, not the desktop.

There are no WM_PAINT message reaching the main form when you move children forms inside the client area.

But there is a WM_PAINT message concerning the client window. We need to intercept it…

Hooking the Client Window WindowProc

We do not have a message handler or a Paint method to override at our disposal. So let's go the Windows way … 

We have to replace the WindowProc of the MDI Client Window, which processes all messages send to it, by our own WindowProc.

It should just do anything done by original one, plus to repaint the client area whenever a WM_PAINT message passes around. It is a bit radical but it works.

Of course, in that case we do not need to repaint the children windows. As we want to avoid code duplication, we'll isolate client painting in a separate routine.

procedure PaintClientWithDesktop(AForm: TForm);
{ handling MDI Client Window }
var
  h: HWND;
begin
  with AForm do begin
    if FormStyle <> fsMDIForm then
      Exit;  { or raise an Exception }
    { paint Client Window with Desktop }
    h:=getDC(ClientHandle);
    if h<>0 then begin
      PaintDesktop(h);
      ReleaseDC(ClientHandle,h);
    end;
  end; { with }
end;


procedure PaintWithDesktop(AForm: TForm);
{ last try : repainting Graphic controls and handling MDI }
var
  i: integer;
begin
  with AForm do begin
    { simply paint Desktop on Canvas }
    PaintDeskTop(Canvas.handle);
    if FormStyle = fsMDIForm then begin
      { need to deal also with Client Window to paint Desktop }
      PaintClientWithDesktop(AForm);
      { repaint also the MDI Children Forms }
      for i:= MDIChildCount-1 downto 0 do
        PaintWithDesktop(MDIChildren[i]);
    end;
    { ask for a repaint of all Graphic controls}
    for i:=0 to ControlCount-1 do
      if Controls[i] is TGraphicControl then begin
        { ******    SPECIAL TRICK : USE WITH CARE    ******     }
        { As Canvas is a protected property in TGraphicControl, }
        { we'll use the simplest descendant to surface it       }
        Controls[i].perform(WM_PAINT, TLabel(Controls[i]).canvas.handle, 0);
      end;
  end; { with }
end;
Hooking technique

Basically, you have to store a pointer to the existing WindowProc, in order to restore it at the end, and also to call it whenever you want the standard behavior. This can be done by overriding the CreateWnd procedure of the MDI Window, to install the new Client WindowProc in, and the DestroyWnd to clean things up.

{-------------------- for MDI only -----------------------------}
procedure TMainForm.CreateWnd;
begin
  inherited CreateWnd;

  if FormStyle <> fsMDIForm then exit;

  { maximum safety ! }
  OldFormProc := nil;
  NewFormProc := nil;
  { get old Client WindowProc }
  OldFormProc := TFarProc(GetWindowLong(ClientHandle, GWL_WNDPROC));
  { install the new proc }
  NewFormProc := MakeObjectInstance(HookWndProc);
  SetWindowLong(ClientHandle, GWL_WNDPROC, LongInt(NewFormProc));
end;

procedure TMainForm.DestroyWnd;
begin
  if Assigned(OldFormProc) then begin
    { install the old proc back }
    SetWindowLong(ClientHandle, GWL_WNDPROC, LongInt(OldFormProc));
    OldFormProc := nil;

    if Assigned(NewFormProc) then
      { some cleaning }
      FreeObjectInstance(NewFormProc);
    NewFormProc := nil;
  end;

  inherited Destroy;
end;


To ease things up we can build a subroutine to execute the standard behavior


procedure TMainForm.HookWndProc(var Message: TMessage);
  procedure DefaultProc;
  begin
    with Message do
      Result := CallWindowProc(OldFormProc, ClientHandle, Msg, wParam, lParam);
  end;
begin
  with Message do
    case Msg of
      WM_PAINT:
        begin
          DefaultProc;
          PaintClientWithDesktop(Self); { <=== that's the added value }
        end;
    else
      DefaultProc;
    end;
end;  { HookWndProc }

{-------------------- That's all Folks ! -----------------------}

Complete MDI Application

I made a complete MDI application with the MDI application wizard.

Just load it into Delphi and run…







